

## Generatore di onde quadre sulla linea RA4 del microcontrollore PIC 16F84A funzionante a 4MHz.

Listato del programma:

```
;***** ONDA QUADRA SULLA LINEA RA4 *****
; fck=4MHz --> 1us/ciclo *
; durata livello alto = durata livello basso = 38 cicli *
; periodo onda quadra: T = 76 us *
; prof. Giuseppe Spalierno - 10 luglio 2012 *
;*****

;Definizione delle direttive per l'assemblatore
PROCESSOR 16F84A
INCLUDE "P16F84A.INC"
ORG 00H

CONTA EQU 0CH

BSF STATUS, RP0 ; Seleziona il banco 1
CLRW ; W=0
MOVWF TRISA ; Pin della porta B tutti di uscita
BCF STATUS, RP0 ; Ritorna al banco 0


INIZIO:
BCF PORTA, RA4 ; RA4=0
CALL RITARDO ; Chiama la routine di ritardo RITARDO
NOP ; Si rendono necessarie due NOP per
NOP ; compensare 2 cicli di GOTO INIZIO
BSF PORTA, RA4 ; RA4=1
CALL RITARDO ; Chiama la routine di ritardo RITARDO
GOTO INIZIO ; Torna a INIZIO

RITARDO:
MOVLW 0AH ; W=10
MOVWF CONTA ; CONTA <-- W=10

LOOP:
DECFSZ CONTA, 1 ; decrementa CONTA e salta la prossima se CONTA=0
GOTO LOOP ; torna a LOOP
RETURN ; ritorna al programma principale
END
```

In figura si mostra l'analisi temporale simulata in ambiente MPLAB 8.86. Il periodo dell'onda quadra rilevata su RA4 vale 76 cicli macchina che corrispondono a 76 us poiché si utilizza un quarzo a 4MHz. Infatti, sapendo che la frequenza del ciclo macchina è  $f_{osc}/4 = 1\text{MHz}$ , si ricava il relativo periodo pari a 1us.

Il livello alto e basso sono uguali tra loro e pari a 38us. La parte iniziale della forma d'onda corrisponde alle prime 4 istruzioni del programma, realizzate una sola volta, all'inizio, che servono a programmare il pin RA4 come linea di uscita.



Eseguiamo il calcolo per verificare la durata.

Si premette che le istruzioni da salto durano 2 cicli macchina e tutte le altre un solo ciclo macchina. La parte di programma che viene continuamente ripetuta è quella che incomincia dall'etichetta INIZIO.

La durata del livello basso vale:

BCF PORTA, RA4 → 1 ciclo macchina  
 CALL RITARDO → 2 cicli macchina (salto) + 33 cicli macchina (durata della subroutine RITARDO)  
 NOP → 1 ciclo macchina  
 NOP → 1 ciclo macchina  
**Totale: 33 + 5 = 38 cicli macchina**

La durata del livello alto vale:

BSF PORTA, RA4 → 1 ciclo macchina  
 CALL RITARDO → 2 cicli macchina (salto) + 33 cicli macchina (durata della subroutine RITARDO)  
 GOTO INIZIO → 2 cicli macchina (salto)  
**Totale: 33 + 5 = 38 cicli macchina**

Verifichiamo che la durata del sottoprogramma RITARDO vale proprio 33 cicli macchina.

MOVLW A → 1 ciclo macchina  
 MOVWF CONTA → 1 ciclo macchina  
 LOOP: DECFSZ CONTA, 1 → 1 ciclo macchina  
 GOTO LOOP → 2 cicli macchina

Avendo inserito in CONTA il valore esadecimale A (cioè il valore decimale 10) le due precedenti istruzioni vengono eseguite  $A-1 = 9$  volte e quindi impegnano  $3 \times 9 = 27$  cicli macchina.

Quando CONTA diventa 0, l'istruzione GOTO LOOP viene saltata per cui la precedente istruzione:

DECFSZ CONTA, 1 diventa una istruzione di salto e quindi → 2 cicli macchina  
 RETURN → 2 cicli macchina  
**Totale: 1 + 1 + 3x9 + 2 + 2 = 33 cicli macchina.**

Generalizzazione della formula.

Detto **n** il valore che inseriremo nell'istruzione MOVLW n che compare dopo l'etichetta RITARDO:

$$TH = TL = 3n + 8$$

Nella seguente tabella si riportano i valori di TL=TH di T=TL+TH e della frequenza in funzione di n.

| n  | TL (us) | TH (us) | T (us) | f (KHz) |
|----|---------|---------|--------|---------|
| 1  | 11      | 11      | 22     | 45,4    |
| 2  | 14      | 14      | 28     | 35,7    |
| 3  | 17      | 17      | 34     | 29,4    |
| 4  | 20      | 20      | 40     | 25      |
| 5  | 23      | 23      | 46     | 21,7    |
| 6  | 26      | 26      | 52     | 19,2    |
| 7  | 29      | 29      | 58     | 17,2    |
| 8  | 32      | 32      | 64     | 15,6    |
| 9  | 35      | 35      | 70     | 14,3    |
| 10 | 38      | 38      | 76     | 13,2    |
| 11 | 41      | 41      | 82     | 12,2    |
| 12 | 44      | 44      | 88     | 11,4    |
| 13 | 47      | 47      | 94     | 10,6    |
| 14 | 50      | 50      | 100    | 10      |

Flow chart del programma

